In[∘]:= Quit[]

In[∘]:= (* Sistema tipo 3-1-3 *)






```
(* Referenciais
    - I (X_i): Inercial (fixo)
    - S1 (x_i^(1)): Móvel 1 (solidário à base giratória)
    - S2 (x_i^(2)): Móvel 2 (solidário à haste giratória)
    - S3 (x_i^(3)): Móvel 3 (solidário ao disco)
*)
(* Matrizes de Transformação *)
```

$$\begin{tikzcd} I \arrow[r, shift left, "\mathbf{T}_\beta"] & S_1 \arrow[l, shift left, "\mathbf{T}_\beta^T"] \arrow[r, shift left, "\mathbf{T}_\theta"] & S_2 \arrow[l, shift left, "\mathbf{T}_\theta^T"] \arrow[r, shift left, "\mathbf{T}_\phi"] & S_3 \arrow[l, shift left, "\mathbf{T}_\phi^T"] \end{tikzcd};$$

$$T\beta = \begin{pmatrix} \mathrm{Cos}[\beta[t]] & \mathrm{Sin}[\beta[t]] & 0 \\ -\mathrm{Sin}[\beta[t]] & \mathrm{Cos}[\beta[t]] & 0 \\ 0 & 0 & 1 \end{pmatrix}; \text{(* Giro em torno de 3 *)}$$

$$T\theta = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \mathrm{Cos}[\theta] & \mathrm{Sin}[\theta] \\ 0 & -\mathrm{Sin}[\theta] & \mathrm{Cos}[\theta] \end{pmatrix}; \text{(* Giro em torno de 1, porém ângulo } \theta \text{ constante *)}$$

$$T\phi = \begin{pmatrix} \text{Cos}[\phi[t]] & \text{Sin}[\phi[t]] & 0 \\ -\text{Sin}[\phi[t]] & \text{Cos}[\phi[t]] & 0 \\ 0 & 0 & 1 \end{pmatrix};$$

In[∘]:=
```
(* Por conveniência, utiliza-se a base S2 para os cálculos *)
(* Velocidades Angulares *)
(* Para determinar as velocidades angulares,
uma estratégia é decompor o vetor em velocidades angulares mais simples: *)
```

$$ {}^{I}_{S_2}\boldsymbol{\omega}^{S_3} = {}^{I}_{S_2}\boldsymbol{\omega}^{S_1} + {}^{S_1}_{S_2}\boldsymbol{\omega}^{S_2} + {}^{S_2}_{S_2}\boldsymbol{\omega}^{S_3} \ ; $$

```
IωIS1 = {0, 0, β'[t]} (* Velocidade angular de S1 em relação a I,
descrita na base I *)
S2ωIS1 = Tθ.Tβ.IωIS1 (* Velocidade angular de S1 em relação a I,
descrita na base S2 *)
S2ωS1S2 = {0, 0, 0}(* Velocidade angular de S2 em relação a S2,
descrita na base S2. Nula pois a haste está fixa na posição θ *)
S2ωS2S3 = {0, 0, ϕ'[t]} (* Velocidade angular de S3 em relação a S2,
descrita na base S2 *)
S2ωIS3 = S2ωIS1 + S2ωS1S2 + S2ωS2S3
(* Velocidade Angular de S3 em relação a I, descrita na base S2 *)
(* É possível determinar também os vetores intermediários, tal que: *)
```

$$ {}^{I}_{S_2}\boldsymbol{\omega}^{S_2} = {}^{I}_{S_2}\boldsymbol{\omega}^{S_1} + {}^{S_1}_{S_2}\boldsymbol{\omega}^{S_2} \ ; \quad {}^{S_1}_{S_2}\boldsymbol{\omega}^{S_3} = {}^{S_1}_{S_2}\boldsymbol{\omega}^{S_2} + {}^{S_2}_{S_2}\boldsymbol{\omega}^{S_3} \ ; $$

```
S2ωIS2 = S2ωIS1 + S2ωS1S2
(* Velocidade angular de S2 em relação a I, descrita na base S2 *)
S2ωS1S3 = S2ωS1S2 + S2ωS2S3
  (* Velocidade angular de S3 em relação a S1, descrita na base S2 *)
```

Out[∘]=
{0, 0, β′[t]}

Out[∘]=
{0, Sin[θ] β′[t], Cos[θ] β′[t]}

Out[∘]=
{0, 0, 0}

Out[∘]=
{0, 0, ϕ′[t]}

Out[∘]=
{0, Sin[θ] β′[t], Cos[θ] β′[t] + ϕ′[t]}

Out[∘]=
{0, Sin[θ] β′[t], Cos[θ] β′[t]}

Out[∘]=
{0, 0, ϕ′[t]}

In[ ]:=
```
(* Acelerações Angulares*)
(* É possível utilizar a seguinte forma
 genérica para determinar as acelerações angulares: *)
```

$${}^{\triangle}_{\bigcirc}\boldsymbol{\alpha}^{\square} = {}^{\triangle}_{\bigcirc}\boldsymbol{\alpha}^{\bigstar} + {}^{\bigstar}_{\bigcirc}\boldsymbol{\alpha}^{\square} + {}^{\triangle}_{\bigcirc}\boldsymbol{\omega}^{\bigstar} \times {}^{\bigstar}_{\bigcirc}\boldsymbol{\omega}^{\square};$$

```
(* Dessa forma, para determinar S2αIS2, faz-se: *)
```

$${}^{I}_{S_2}\boldsymbol{\alpha}^{S_2} = {}^{I}_{S_2}\boldsymbol{\alpha}^{S_1} + {}^{S_1}_{S_2}\boldsymbol{\alpha}^{S_2} + {}^{I}_{S_2}\boldsymbol{\omega}^{S_1} \times {}^{S_1}_{S_2}\boldsymbol{\omega}^{S_2};$$

```
S2αIS1 = D[S2ωIS1, t] (* Aceleração angular de S1 em relação a I,
descrita na base S2 *)
S2αS1S2 = {0, 0, 0} (* Aceleração angular de S2 em relação a S1,
descrita na base S2. Nula pois haste está fixa na posição θ *)
S2αIS2Cross = Cross[S2ωIS1, S2ωS1S2]
(* Produto vetorial necessário para calcular S2αIS2 *)
S2αIS2 = S2αIS1 + S2αS1S2 + S2αIS2Cross
(* Aceleração angular de S2 em relação a I, descrito na base S2 *)
(* De forma similar, para determinar S2αIS3, faz-se: *)
```

$${}^{I}_{S_2}\boldsymbol{\alpha}^{S_3} = {}^{I}_{S_2}\boldsymbol{\alpha}^{S_2} + {}^{S_2}_{S_2}\boldsymbol{\alpha}^{S_3} + {}^{I}_{S_2}\boldsymbol{\omega}^{S_2} \times {}^{S_2}_{S_2}\boldsymbol{\omega}^{S_3};$$

```
S2αS2S3 = D[S2ωS2S3, t] (* Aceleração angular de S3 em relação a S2,
descrita na base S2 *)
S2αIS3Cross = Cross[S2ωIS1, S2ωS2S3]
(* Produto vetorial necessário para calcular S2αIS3 *)
S2αIS3 = S2αIS1 + S2αS2S3 + S2αIS3Cross
  (* Aceleração angular de S3 em relação a I, descrita na base S2 *)
```

Out[ ]=
```
{0, Sin[θ] β''[t], Cos[θ] β''[t]}
```

Out[ ]=
```
{0, 0, 0}
```

Out[ ]=
```
{0, 0, 0}
```

Out[ ]=
```
{0, Sin[θ] β''[t], Cos[θ] β''[t]}
```

Out[ ]=
```
{0, 0, ϕ''[t]}
```

Out[ ]=
```
{Sin[θ] β'[t] ϕ'[t], 0, 0}
```

Out[ ]=
```
{Sin[θ] β'[t] ϕ'[t], Sin[θ] β''[t], Cos[θ] β''[t] + ϕ''[t]}
```

In[ ]:=
```
(* Calculo do Vetor Posição Absoluta *)
(* A o vetor posição absoluta de P pode ser
 decomposto em partes mais simples como mostrado abaixo: *)
```

$$ {}^{I}_{S_2}\mathbf{p}^{P} = {}^{I}_{S_2}\mathbf{p}^{S_1} + {}^{S_1}_{S_2}\mathbf{p}^{S_2} + {}^{S_2}_{S_2}\mathbf{p}^{S_3} + {}^{S_3}_{S_2}\mathbf{p}^{P} \;; $$

```
IpIS1 = {0, 0, h} (* Posição de S1 em relação a I, descrita na base I *)
S1pIS1 = Tβ.IpIS1 (* Posição de S1 em relação a I, descrita na base S1 *)
S2pIS1 = Tθ.S1pIS1 (* Posição de S1 em relação a I, descrita na base S2 *)
S2pS1S2 = {0, 0, 0} (* Posição de S2 em relação a S1,
descrita na base S2. Nula pois ambas as
 origens dos dois referenciais são coincidentes *)
S2pS2S3 = {0, 0, L} (* Posição de S2 em relação a S3, descrita na base S2 *)
S3pS3P = {R, 0, 0} (* Posição da partícula P em relação ao referencial S3,
descrita na base S3 *)
S2pS3P = Transpose[Tϕ].S3pS3P
(* Posição da partícula P em relação ao referencial S3, descrita na base S2 *)
S2pIP = S2pIS1 + S2pS1S2 + S2pS2S3 + S2pS3P
  (* Posição absoluta da partícula P, descrita na base S2 *)
```

Out[ ]=
{0, 0, h}

Out[ ]=
{0, 0, h}

Out[ ]=
{0, h Sin[θ], h Cos[θ]}

Out[ ]=
{0, 0, 0}

Out[ ]=
{0, 0, L}

Out[ ]=
{R, 0, 0}

Out[ ]=
{R Cos[ϕ[t]], R Sin[ϕ[t]], 0}

Out[ ]=
{R Cos[ϕ[t]], h Sin[θ] + R Sin[ϕ[t]], L + h Cos[θ]}

In[ ]:=
```
(* Velocidade Absoluta de P *)
(* Em casos envolvendo mais de 1 referencial móvel,
é necessário aplicar o teorema cinemático de
 forma recursiva para determinar todos os termos. Isto é,
divide-se o problema em formas mais simples. Para isso, é possível escrever
 o teorema cinemático para velocidade da seguinte forma genérica: *)
```

$$ {}^{\triangle}_{\bigcirc}\mathbf{v}^{\square} = {}^{\triangle}_{\bigcirc}\mathbf{v}^{\bigstar} + {}^{\bigstar}_{\bigcirc}\mathbf{v}^{\square} + {}^{\triangle}_{\bigcirc}\omega^{\bigstar} \times {}^{\bigstar}_{\bigcirc}\mathbf{p}^{\square} \;; $$

```
(* Para achar S2vIP, escolhe-se utilizar o referencial S3 no teorema,
isto é, * = S3. O que resulta na seguinte forma do teorema cinemático: *)
```

$$ {}^{I}_{S_2}\mathbf{v}^{P} = {}^{I}_{S_2}\mathbf{v}^{S_3} + {}^{S_3}_{S_2}\mathbf{v}^{P} + {}^{I}_{S_2}\boldsymbol{\omega}^{S_3} \times {}^{S_3}_{S_2}\mathbf{p}^{P} \;; $$

```
(* Porém, para achar S2vIS3 - primeiro termo da equação acima,
é necessário aplicar novamente o teorema
 cinemático. Utilizando agora o referencial S2, isto é *=S2,
resulta na seguinte forma do teorema cinemático auxiliar: *)
```

$$ {}^{I}_{S_2}\mathbf{v}^{S_3} = {}^{I}_{S_2}\mathbf{v}^{S_2} + {}^{S_2}_{S_2}\mathbf{v}^{S_3} + {}^{I}_{S_2}\boldsymbol{\omega}^{S_2} \times {}^{S_2}_{S_2}\mathbf{p}^{S_3} \;; $$

```
(* Com isso,
definem-se os termos necessários para o cálculo da velocidade abosluta de P *)
S2vIS2 = {0, 0, 0} (* Velocidade de S2 em relação a I,
descrita na base S2. Nula pois não há variação de posição entre S2 e I. *)
S2vS2S3 = {0, 0, 0} (* Velocidade de S3 em relação a S2, descrita na base
 S2. Nula pois não há variação de posição de S3 em relação a S2. *)
S2vIS3Cross = Cross[S2ωIS2, S2pS2S3]
(* Velocidade tangencial: Produto vetorial necessário para calcular S2vIS3 *)
S2vIS3 = S2vIS2 + S2vS2S3 + S2vIS3Cross
(* Velocidade de S3 em relação a I, descrita na base S2 *)
S2vS3P = {0, 0, 0} (* Velocidade de P em relação a S3,
descrita na base S2. Nula pois não há velocidade relativa entre P e S3 *)

S2vIPCross = Cross[S2ωIS3, S2pS3P]
(* Velocidade Tangencial: Produto vetorial necessário para calcular S2vIP *)
S2vIP = S2vIS3 + S2vS3P + S2vIPCross // FullSimplify
(* Velocidade absoluta de P, descrita na base S2 *)
S1vIP = Transpose[Tθ].S2vIP // FullSimplify
(* Velocidade absoluta de P, descrita na base S1 *)
S3vIP = Tϕ.S2vIP // FullSimplify
(* Velocidade absoluta de P, descrita na base S1 *)
IvIP = Transpose[Tβ].S1vIP // FullSimplify
  (* Velocidade absoluta de P, descrita na base I *)
```

Out[ ]=
```
{0, 0, 0}
```

Out[ ]=
{0, 0, 0}

Out[ ]=
{L Sin[θ] β′[t], 0, 0}

Out[ ]=
{L Sin[θ] β′[t], 0, 0}

Out[ ]=
{0, 0, 0}

Out[ ]=
{-R Cos[θ] Sin[ϕ[t]] β′[t] - R Sin[ϕ[t]] ϕ′[t],
R Cos[θ] Cos[ϕ[t]] β′[t] + R Cos[ϕ[t]] ϕ′[t], -R Cos[ϕ[t]] Sin[θ] β′[t]}

Out[ ]=
{L Sin[θ] β′[t] - R Sin[ϕ[t]] (Cos[θ] β′[t] + ϕ′[t]),
R Cos[ϕ[t]] (Cos[θ] β′[t] + ϕ′[t]), -R Cos[ϕ[t]] Sin[θ] β′[t]}

Out[ ]=
{L Sin[θ] β′[t] - R Sin[ϕ[t]] (Cos[θ] β′[t] + ϕ′[t]),
R Cos[ϕ[t]] (β′[t] + Cos[θ] ϕ′[t]), R Cos[ϕ[t]] Sin[θ] ϕ′[t]}

Out[ ]=
{L Cos[ϕ[t]] Sin[θ] β′[t],
(R Cos[θ] - L Sin[θ] Sin[ϕ[t]]) β′[t] + R ϕ′[t], -R Cos[ϕ[t]] Sin[θ] β′[t]}

Out[ ]=
{-R Cos[ϕ[t]] Sin[β[t]] (β′[t] + Cos[θ] ϕ′[t]) +
Cos[β[t]] (L Sin[θ] β′[t] - R Sin[ϕ[t]] (Cos[θ] β′[t] + ϕ′[t])),
R Cos[β[t]] Cos[ϕ[t]] (β′[t] + Cos[θ] ϕ′[t]) + Sin[β[t]]
(L Sin[θ] β′[t] - R Sin[ϕ[t]] (Cos[θ] β′[t] + ϕ′[t])), R Cos[ϕ[t]] Sin[θ] ϕ′[t]}

In[ ]:=
```
(* Vetor aceleração absoluta de P *)
(* Como anteriormente,
o teorema cinemático pode ser escrito na forma genérica como: *)
```

$$\overset{\triangle}{\underset{\bigcirc}{}}\mathbf{a}^{\square} = \overset{\triangle}{\underset{\bigcirc}{}}\mathbf{a}^{\bigstar} + \overset{\bigstar}{\underset{\bigcirc}{}}\mathbf{a}^{\square} + \overset{\triangle}{\underset{\bigcirc}{}}\alpha^{\bigstar} \times \overset{\bigstar}{\underset{\bigcirc}{}}\mathbf{p}^{\square} + \overset{\triangle}{\underset{\bigcirc}{}}\omega^{\bigstar} \times \left(\overset{\triangle}{\underset{\bigcirc}{}}\omega^{\bigstar} \times \overset{\bigstar}{\underset{\bigcirc}{}}\mathbf{p}^{\square}\right) + 2\overset{\triangle}{\underset{\bigcirc}{}}\omega^{\bigstar} \times \overset{\bigstar}{\underset{\bigcirc}{}}\mathbf{v}^{\square}$$

```
(* Dessa forma, para achar S2aIP,
escolhe-se utilizar o referencial S3 no teorema, isto é, * = S3. Com isso,
o teorema cinemático pode ser escrito da seguinte forma: *)
```

$${}^{I}_{S_2}\mathbf{a}^{P} = {}^{I}_{S_2}\mathbf{a}^{S_3} + {}^{S_3}_{S_2}\mathbf{a}^{P} + {}^{I}_{S_2}\boldsymbol{\alpha}^{S_3} \times {}^{S_3}_{S_2}\mathbf{p}^{P} + {}^{I}_{S_2}\boldsymbol{\omega}^{S_3} \times \left({}^{I}_{S_2}\boldsymbol{\omega}^{S_3} \times {}^{S_3}_{S_2}\mathbf{p}^{P}\right) + 2{}^{I}_{S_2}\boldsymbol{\omega}^{S_3}$$

```
(* Para determinar S1aIS3 é necessário aplicar outro teorema
 cinemático. Utilizando agora o referencial S2, isto é * = S2,
o teorema cinemático auxiliar é descrito da seguinte forma: *)
```

$${}^{I}_{S_2}\mathbf{a}^{S_3} = {}^{I}_{S_2}\mathbf{a}^{S_2} + {}^{S_2}_{S_2}\mathbf{a}^{S_3} + {}^{I}_{S_2}\boldsymbol{\alpha}^{S_2} \times {}^{S_2}_{S_2}\mathbf{p}^{S_3} + {}^{I}_{S_2}\boldsymbol{\omega}^{S_2} \times \left({}^{I}_{S_2}\boldsymbol{\omega}^{S_2} \times {}^{S_2}_{S_2}\mathbf{p}^{S_3}\right) + 2{}^{I}_{S_2}\boldsymbol{\omega}^{S_2}$$

```
(* Com isso,
definem-se os termos necessários para o cálculo da aceleração absoluta de P: *)
S2aIS2 = {0, 0, 0} (* Aceleração de S2 em relação a I,
descrita na base S2. Nula pois não há
```

```
 variação do módulo da posição de S2 em relação a I *)
S2aS2S3 = {0, 0, 0} (* Aceleração de S3 em relação a S2. Nula pois
 não há variação do módulo da posição de S3 em relação a S2 *)
S2aIS3Cross1 = Cross[S2αIS2, S2pS2S3]
(* Aceleração tangencial. Primeiro produto vetorial para determinar S2aIS3 *)
S2aIS3Cross2 = Cross[S2ωIS2, Cross[S2ωIS2, S2pS2S3]]
(* Aceleração normal. Segundo produto vetorial para determinar S2aIS3 *)
S2aIS3Cross3 = 2 * Cross[S2ωIS2, S2vS2S3]
(* Aceleração de Coriolis. Terceiro produto vetorial para determinar S2aIS3 *)
S2aIS3 = S2aIS2 + S2aS2S3 + S2aIS3Cross1 + S2aIS3Cross2 + S2aIS3Cross3
(* Aceleração de S3 em relação a I, descrita na base S2 *)
S2aS3P = {0, 0, 0} (* Aceleração de P em relação a S3,
descrita na base S2. Nula pois não há variação
 do módulo da velocidade de P em relação a S3. *)
S2aIPCross1 = Cross[S2αIS3, S2pS3P] // FullSimplify
(* Aceleração tangencial. Primeiro produto
 vetorial necessário para determinar S2aIP *)
S2aIPCross2 = Cross[S2ωIS3, Cross[S2ωIS3, S2pS3P]] // FullSimplify
(* Aceleração normal. Segundo produto
 vetorial necessário para determinar S2aIP *)
S2aIPCross3 = 2 * Cross[S2ωIS3, S2vS3P] // FullSimplify
(* Aceleração de Coriolis. Terceiro produto
 vetorial necessário para determinar S2aIP *)
S2aIP = S2aIS3 + S2aS3P + S2aIPCross1 + S2aIPCross2 + S2aIPCross3 // FullSimplify
(* Aceleração absoluta de P, descrita na base S2. *)
S1aIP = Transpose[Tθ].S2aIP // FullSimplify
(* Aceleração absoluta de P, descrita na base S1 *)
S3aIP = Tϕ.S2aIP // FullSimplify
(* Aceleração absoluta de P, descrita na base S1 *)
IaIP = Transpose[Tβ].S1aIP // FullSimplify
  (* Aceleração absoluta de P, descrita na base I *)
```

Out[ ]=

{0, 0, 0}

Out[ ]=

{0, 0, 0}

Out[ ]=

$\{L\,\mathrm{Sin}[\theta]\,\beta''[t],\,0,\,0\}$

Out[ ]=

$\{0,\,L\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,\beta'[t]^2,\,-L\,\mathrm{Sin}[\theta]^2\,\beta'[t]^2\}$

Out[ ]=

{0, 0, 0}

Out[ ]=

$\{L\,\mathrm{Sin}[\theta]\,\beta''[t],\,L\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,\beta'[t]^2,\,-L\,\mathrm{Sin}[\theta]^2\,\beta'[t]^2\}$

Out[ ]=

{0, 0, 0}

Out[ ]=

```
{-R Sin[φ[t]] (Cos[θ] β''[t] + φ''[t]), R Cos[φ[t]] (Cos[θ] β''[t] + φ''[t]),
 R Sin[θ] (Sin[φ[t]] β'[t] φ'[t] - Cos[φ[t]] β''[t])}
```

Out[ ]=

```
{-R Cos[φ[t]] (β'[t]^2 + 2 Cos[θ] β'[t] φ'[t] + φ'[t]^2),
 -R Sin[φ[t]] (Cos[θ] β'[t] + φ'[t])^2, R Sin[θ] Sin[φ[t]] β'[t] (Cos[θ] β'[t] + φ'[t])}
```

Out[ ]=

```
{0, 0, 0}
```

Out[ ]=

```
{-R Cos[φ[t]] (β'[t]^2 + 2 Cos[θ] β'[t] φ'[t] + φ'[t]^2) +
   L Sin[θ] β''[t] - R Sin[φ[t]] (Cos[θ] β''[t] + φ''[t]),
 L Cos[θ] Sin[θ] β'[t]^2 - R Sin[φ[t]] (Cos[θ] β'[t] + φ'[t])^2 +
   R Cos[φ[t]] (Cos[θ] β''[t] + φ''[t]), Sin[θ] ((-L Sin[θ] + R Cos[θ] Sin[φ[t]]) β'[t]^2 +
     2 R Sin[φ[t]] β'[t] φ'[t] - R Cos[φ[t]] β''[t])}
```

Out[ ]=

```
{-R Cos[φ[t]] (β'[t]^2 + 2 Cos[θ] β'[t] φ'[t] + φ'[t]^2) +
   L Sin[θ] β''[t] - R Sin[φ[t]] (Cos[θ] β''[t] + φ''[t]),
 L Sin[θ] β'[t]^2 - R Sin[φ[t]] (2 β'[t] φ'[t] + Cos[θ] (β'[t]^2 + φ'[t]^2)) +
   R Cos[φ[t]] (β''[t] + Cos[θ] φ''[t]), R Sin[θ] (-Sin[φ[t]] φ'[t]^2 + Cos[φ[t]] φ''[t])}
```

Out[ ]=

```
{-((R Cos[φ[t]]^2 + Cos[θ] Sin[φ[t]] (-L Sin[θ] + R Cos[θ] Sin[φ[t]])) β'[t]^2) -
   2 R Cos[θ] β'[t] φ'[t] - R φ'[t]^2 + L Cos[φ[t]] Sin[θ] β''[t],
 Cos[φ[t]] Sin[θ] (L Cos[θ] + R Sin[θ] Sin[φ[t]]) β'[t]^2 +
   (R Cos[θ] - L Sin[θ] Sin[φ[t]]) β''[t] + R φ''[t],
 Sin[θ] ((-L Sin[θ] + R Cos[θ] Sin[φ[t]]) β'[t]^2 +
     2 R Sin[φ[t]] β'[t] φ'[t] - R Cos[φ[t]] β''[t])}
```

Out[ ]=

```
{Cos[β[t]] (-R Cos[φ[t]] (β'[t]^2 + 2 Cos[θ] β'[t] φ'[t] + φ'[t]^2) +
       L Sin[θ] β''[t] - R Sin[φ[t]] (Cos[θ] β''[t] + φ''[t])) -
   Sin[β[t]] (L Sin[θ] β'[t]^2 - R Sin[φ[t]] (2 β'[t] φ'[t] + Cos[θ] (β'[t]^2 + φ'[t]^2)) +
       R Cos[φ[t]] (β''[t] + Cos[θ] φ''[t])),
 Sin[β[t]] (-R Cos[φ[t]] (β'[t]^2 + 2 Cos[θ] β'[t] φ'[t] + φ'[t]^2) +
       L Sin[θ] β''[t] - R Sin[φ[t]] (Cos[θ] β''[t] + φ''[t])) +
   Cos[β[t]] (L Sin[θ] β'[t]^2 - R Sin[φ[t]] (2 β'[t] φ'[t] + Cos[θ] (β'[t]^2 + φ'[t]^2)) +
       R Cos[φ[t]] (β''[t] + Cos[θ] φ''[t])),
 R Sin[θ] (-Sin[φ[t]] φ'[t]^2 + Cos[φ[t]] φ''[t])}
```